Problem:
Score : 500 points

Problem Statement

We have a two-dimensional grid with H \times W squares. There are M targets to destroy in this grid - the position of the i-th target is \left(h_i, w_i \right).

Takahashi will choose one square in this grid, place a bomb there, and ignite it. The bomb will destroy all targets that are in the row or the column where the bomb is placed. It is possible to place the bomb at a square with a target.

Takahashi is trying to maximize the number of targets to destroy. Find the maximum number of targets that can be destroyed.

Constraints

All values in input are integers.

1 \leq H, W \leq 3 \times 10^5

1 \leq M \leq \min\left(H\times W, 3 \times 10^5\right)

1 \leq h_i \leq H

1 \leq w_i \leq W

\left(h_i, w_i\right) \neq \left(h_j, w_j\right) \left(i \neq j\right)

Input

Input is given from Standard Input in the following format:

H W M
h_1 w_1
\vdots
h_M w_M

Output

Print the answer.

Sample Input 1

2 3 3
2 2
1 1
1 3

Sample Output 1

3

We can destroy all the targets by placing the bomb at \left(1, 2\right).

Sample Input 2

3 3 4
3 3
3 1
1 1
1 2

Sample Output 2

3

Sample Input 3

5 5 10
2 5
4 3
2 3
5 5
2 2
5 4
5 3
5 1
3 5
1 4

Sample Output 3

6

Solution in Perl (5.26.1) (judge verdict: Accepted)

use strict;
use warnings;
use utf8;

use feature qw(say);
use Data::Dumper;

my $k = <STDIN>;
chomp $k;
my ($h, $w, $m) = split / /, $k;

my $h_list = [];
my $h_count_mapping = +{};
my $w_count_mapping = +{};
while (my $input = <STDIN>) {
    chomp $input;
    last unless $input;
    my ($h_n, $w_n) = split / /, $input;
    $h_count_mapping->{$h_n} = ($h_count_mapping->{$h_n} || 0) + 1;
    $w_count_mapping->{$w_n} = ($w_count_mapping->{$w_n} || 0) + 1;

    push @$h_list, [ $h_n, $w_n, ];
}

my ($max_h, $max_h_keys) = max_in_mapping($h_count_mapping);
my ($max_w, $max_w_keys) = max_in_mapping($w_count_mapping);

my $max_h_mapping = +{map { $_ => 1 } @$max_h_keys};
my $max_w_mapping = +{map { $_ => 1 } @$max_w_keys};

my $duplicated_count = 0;
for my $one (@$h_list) {
    my $one_h = $one->[0];
    my $one_w = $one->[1];

    if ($max_h_mapping->{$one_h} && $max_w_mapping->{$one_w}) {
        $duplicated_count++;
    }
}
my $candidate_count = scalar @$max_h_keys * scalar @$max_w_keys;
my $has_duplicated = $candidate_count - $duplicated_count > 0 ? 0 : 1;
my $total_count = $max_h + $max_w - $has_duplicated;
say $total_count;


sub max_in_mapping {
    my $mapping = shift;

    my $max = 0;
    my $max_keys;

    for my $key (keys %$mapping) {
        my $count = $mapping->{$key};
        my $diff = $count - $max;
        if ($diff > 0) {
            $max = $count;
            $max_keys = [$key];
        } elsif ($diff == 0) {
            push @$max_keys, $key;
        }
    }
    return ($max, $max_keys);
}